Math 2650 - Linear Differential Equations

A. J. Meir

Forced Undamped Oscillators

We consider equations of the type

$$m\frac{d^2x}{dt^2} + kx = F\cos(\omega t).$$

For example take $m = 1$, $k = 9$, and $F = 4$ (the roots of the characteristic polynomial are $3\,I$ and $-3\,I$)

$$\frac{d^2x}{dt^2} + 9\,x = 4\cos(\omega t)$$

with initial conditions

$$x(0) = 0 \quad \text{and} \quad \frac{dx}{dt}(0)=0.$$

```
> restart:with(DEtools):
> dsolve({D(D(x))(t)+9*x(t)=4*cos(omega*t),x(0)=0,D(x)(0)=0},x
  (t));
```

$$x(t) = \frac{4\cos(3\,t)}{-9+\omega^2} - \frac{4\cos(\omega\,t)}{-9+\omega^2}$$

trig identities

That is given a periodic function of the form

$$f(t) = a\cos(\omega t) + b\sin(\omega t)$$

we can write it as

$$f(t) = A\cos(\omega t - \delta)$$

where

$$A = \sqrt{a^2 + b^2}$$

and

$$\delta = \arctan\left(\frac{b}{a}\right)$$

(note this last equation has two solutions which differ by π, you must chose the correct solution).

Also recall the two trigonometric identities

$$\cos(\alpha) - \cos(\beta) = -2\sin\left(\frac{\alpha+\beta}{2}\right)\sin\left(\frac{\alpha-\beta}{2}\right)$$

and

$$\sin(\alpha) - \sin(\beta) = 2\cos\left(\frac{\alpha+\beta}{2}\right)\sin\left(\frac{\alpha-\beta}{2}\right)$$

using these identities we can often rewrite periodic functions to better see the phenomenon of beats.

```
> sol(t):=-8/(-9+omega^2)*sin((3-omega)*t/2)*sin((3+omega)*t/2)
;
```

$$sol(t) := -\frac{8\sin\left(\frac{1}{2}(3-\omega)t\right)\sin\left(\frac{1}{2}(3+\omega)t\right)}{-9+\omega^2}$$

```
> env(t):=-8/(-9+omega^2)*sin((3-omega)*t/2);
```

$$env(t) := -\frac{8\sin\left(\frac{1}{2}(3-\omega)t\right)}{-9+\omega^2}$$

```
> omega:=1:plot([sol(t),env(t),-env(t)],t=0..20,color=[black,
red,red]);
```




```
> omega:=2:plot([sol(t),env(t),-env(t)],t=0..20,color=[black,
red,red]);
```




```
> omega:=5/2:plot([sol(t),env(t),-env(t)],t=0..20,color=[black,
  red,red]);
```




```
> omega:=1.75*sqrt(2):plot([sol(t),env(t),-env(t)],t=0..20,
  color=[black,red,red]);
```




```
> omega:=29/10:plot([sol(t),env(t),-env(t)],t=0..50,color=
  [black,red,red]);
```



For $\omega = 3$ we get

```
> dsolve({D(D(x))(t)+9*x(t)=4*cos(3*t),x(0)=0,D(x)(0)=0},x(t));
```

$$x(t) = \frac{2}{3}\sin(3\,t)\,t$$

```
> sol(t):=2/3*sin(3*t)*t;
```

$$sol(t) := \frac{2}{3}\sin(3\,t)\,t$$

```
> env(t):=2/3*t;
```

$$env(t) := \frac{2}{3}\,t$$

```
> plot([sol(t),env(t),-env(t)],t=0..25,color=[black,red,red]);
```




```
>
```

Forced Damped Oscillators

We consider equations of the type

$$m\frac{d^2 x}{dt^2} + c\frac{dx}{dt} + kx = F\cos(\omega\, t)$$

For example assume that $m=1$, $k=1$, $c=\frac{1}{8}$, and $F=3$ (the roots of the characterisatic polynomial are $\frac{-1 + I\sqrt{255}}{\sqrt{256}}$ and $\frac{-1 - I\sqrt{255}}{\sqrt{256}}$).

We now in addition assume that $\omega = 0.3$ and solve the initial value problem with initial conditions

$$x(0)=0 \qquad \frac{dx}{dt}(0)=0$$

```
> restart:with(DEtools):with(plots):
Warning, the name changecoords has been redefined
> de:=diff(x(t),t,t)+1/8*diff(x(t),t)+x(t)=3*cos(omega*t);
```

$$de := \frac{d^2}{dt^2}\,x(t) + \frac{1}{8}\,\frac{d}{dt}\,x(t) + x(t) = 3\cos(\omega\, t)$$

```
> dsolve({subs(omega=3/10,de),x(0)=0,D(x)(0)=0},x(t));
```

$$x(t) = -\frac{34880}{2256257}\,e^{-\frac{1}{16}\,t}\sin\left(\frac{1}{16}\sqrt{255}\,t\right)\sqrt{255} - \frac{436800}{132721}\,e^{-\frac{1}{16}\,t}\cos\left(\frac{1}{16}\sqrt{255}\,t\right)$$

$$+\frac{18000}{132721}\sin\left(\frac{3}{10}t\right)+\frac{436800}{132721}\cos\left(\frac{3}{10}t\right)$$

We now identify the steady state (the term that persists in *t*) and the transient the term that decays in *t*.

```
> sol:=rhs(%);
```

$$sol := -\frac{34880}{2256257}e^{-\frac{1}{16}t}\sin\left(\frac{1}{16}\sqrt{255}\,t\right)\sqrt{255}-\frac{436800}{132721}e^{-\frac{1}{16}t}\cos\left(\frac{1}{16}\sqrt{255}\,t\right)+\frac{18000}{132721}\sin\left(\frac{3}{10}t\right)+\frac{436800}{132721}\cos\left(\frac{3}{10}t\right)$$

```
> transient:=select(has,sol,exp);
```

$$transient := -\frac{34880}{2256257}e^{-\frac{1}{16}t}\sin\left(\frac{1}{16}\sqrt{255}\,t\right)\sqrt{255}-\frac{436800}{132721}e^{-\frac{1}{16}t}\cos\left(\frac{1}{16}\sqrt{255}\,t\right)$$

```
> steady:=remove(has,sol,exp);
```

$$steady := \frac{18000}{132721}\sin\left(\frac{3}{10}t\right)+\frac{436800}{132721}\cos\left(\frac{3}{10}t\right)$$

We now plot the solution, steady state and transient on the same coordinate system.

```
> plot({sol,transient,steady},
  t=0..40,color=[magenta,red,blue]);
```



and

```
> plot({sol,transient,steady},
  t=0..80,color=[magenta,red,blue]);
```



Note that the solution "converges" to the staedy state as $t \to \infty$. Also note that the steady state is periodic with period $\frac{20\,\pi}{3}$ which is the same as the period of the forcing function. Moreover, we observe that the steady state is not in phase with the forcing. To determine the phase difference between the steady state and forcing, we write the steady state in phase-amplitude form.

```
> s0:=simplify(subs(t=0,steady));
```

$$s0 := \frac{436800}{132721}$$

```
> phase_amp:=A*cos(3/10*t-delta);
```

$$phase\_amp := A\cos\left(-\frac{3}{10}\,t + \delta\right)$$

```
> p0:=simplify(subs(t=0, phase_amp));
```

$$p0 := A\cos(\delta)$$

```
> s1:=simplify(subs(t=0,diff(steady,t)));
```

$$s1 := \frac{5400}{132721}$$

```
> p1:=simplify(subs(t=0, diff(phase_amp,t)));
```

$$p1 := \frac{3}{10}\,A\sin(\delta)$$

```
> solve({p0=s0,p1=s1},{A,delta});
```

$$\{\delta = \arctan(15\,\mathrm{RootOf}(-1 + 132721\,\_Z^2, label=\_L7), 364\,\mathrm{RootOf}(-1 + 132721\,\_Z^2, label=\_L7)), A = 1200\,\mathrm{RootOf}(-1 + 132721\,\_Z^2, label=\_L7)\}$$

```
> allvalues(%);
```

$$\left\{\delta = \arctan\left(\frac{15}{364}\right), A = \frac{1200}{132721}\sqrt{132721}\right\}, \left\{A = -\frac{1200}{132721}\sqrt{132721}, \delta = \arctan\left(\frac{15}{364}\right) - \pi\right\}$$

```
> evalf(%);
```

$$\{A=3.293907684,\ \delta=0.04118548851\},\ \{\delta=-3.100407165,\ A=-3.293907684\}$$

The steady state is displaced in phase by about 0.0412 from the forcing.

Next we show that the amplitude and phase of the steady state are independent of the initial conditions (hence the initial conditions affect only the transient). To do this we find a general solution of the equation and show that $x(0)=x_0$ and $\frac{dx}{dt}(0)=x_1$ appear only in the transient terms.

```
> dsolve({subs(omega=3/10,de),x(0)=x_0,D(x)(0)=x_1},x(t));
```

$$x(t) = \frac{1}{33843855} e^{-\frac{1}{16} t} \sin\left(\frac{1}{16}\sqrt{255}\, t\right)\sqrt{255}\,(2123536\, x\_1 - 523200 + 132721\, x\_0) + e^{-\frac{1}{16} t}\cos\left(\frac{1}{16}\sqrt{255}\, t\right)\left(x\_0 - \frac{436800}{132721}\right) + \frac{18000}{132721}\sin\left(\frac{3}{10} t\right) + \frac{436800}{132721}\cos\left(\frac{3}{10} t\right)$$

```
> combine(%,trig):
> collect(%,exp);
```

$$x(t) = \left(\frac{16}{255}\sin\left(\frac{1}{16}\sqrt{255}\, t\right)\sqrt{255}\, x\_1 - \frac{34880}{2256257}\sin\left(\frac{1}{16}\sqrt{255}\, t\right)\sqrt{255} + \frac{1}{255}\sin\left(\frac{1}{16}\sqrt{255}\, t\right)\sqrt{255}\, x\_0 + \cos\left(\frac{1}{16}\sqrt{255}\, t\right) x\_0 - \frac{436800}{132721}\cos\left(\frac{1}{16}\sqrt{255}\, t\right)\right) e^{-\frac{1}{16} t} + \frac{18000}{132721}\sin\left(\frac{3}{10} t\right) + \frac{436800}{132721}\cos\left(\frac{3}{10} t\right)$$

We now consider the general equation where ω is a fixed positive constant. We solve the equation with that initial conditions $x(0)=x_0$ and $\frac{dx}{dt}(0)=x_1$. We again identify the steady state and show it is independent of x_0 and x_1.

```
> de;
```

$$\frac{d^2}{dt^2} x(t) + \frac{1}{8}\frac{d}{dt} x(t) + x(t) = 3\cos(\omega t)$$

```
> dsolve({de,x(0)=x_0,D(x)(0)=x_1},x(t));
```

$$x(t) = \frac{1}{255}\,\frac{1}{64 - 127\,\omega^2 + 64\,\omega^4}\left(e^{-\frac{1}{16} t}\sin\left(\frac{1}{16}\sqrt{255}\, t\right)\sqrt{255}\,(1024\, x\_1\,\omega^4 - 192 + 64\, x\_0\,\omega^4 - 2032\, x\_1\,\omega^2 - 192\,\omega^2 - 127\, x\_0\,\omega^2 + 1024\, x\_1 + 64\, x\_0)\right)$$

$$+\frac{e^{-\frac{1}{16}t}\cos\left(\frac{1}{16}\sqrt{255}\,t\right)\left(64\,x\_0-127\,x\_0\,\omega^2+64\,x\_0\,\omega^4-192+192\,\omega^2\right)}{64-127\,\omega^2+64\,\omega^4}$$

$$+\frac{24\sin(\omega\,t)\,\omega+192\cos(\omega\,t)-192\,\omega^2\cos(\omega\,t)}{64-127\,\omega^2+64\,\omega^4}$$

```
> combine(%,trig):
> gen_sol:=collect(rhs(%),exp);
```

$$gen\_sol := \frac{1}{16320-32385\,\omega^2+16320\,\omega^4}\Bigg(\Bigg(-2032\sin\left(\frac{1}{16}\sqrt{255}\,t\right)\sqrt{255}\,x\_1\,\omega^2$$

$$-192\sin\left(\frac{1}{16}\sqrt{255}\,t\right)\sqrt{255}\,\omega^2-127\sin\left(\frac{1}{16}\sqrt{255}\,t\right)\sqrt{255}\,x\_0\,\omega^2$$

$$+1024\sin\left(\frac{1}{16}\sqrt{255}\,t\right)\sqrt{255}\,x\_1+64\sin\left(\frac{1}{16}\sqrt{255}\,t\right)\sqrt{255}\,x\_0$$

$$+16320\cos\left(\frac{1}{16}\sqrt{255}\,t\right)x\_0-32385\cos\left(\frac{1}{16}\sqrt{255}\,t\right)x\_0\,\omega^2$$

$$+16320\cos\left(\frac{1}{16}\sqrt{255}\,t\right)x\_0\,\omega^4-48960\cos\left(\frac{1}{16}\sqrt{255}\,t\right)$$

$$+48960\cos\left(\frac{1}{16}\sqrt{255}\,t\right)\omega^2+1024\sin\left(\frac{1}{16}\sqrt{255}\,t\right)\sqrt{255}\,x\_1\,\omega^4$$

$$-192\sin\left(\frac{1}{16}\sqrt{255}\,t\right)\sqrt{255}+64\sin\left(\frac{1}{16}\sqrt{255}\,t\right)\sqrt{255}\,x\_0\,\omega^4\Bigg)e^{-\frac{1}{16}t}\Bigg)$$

$$+\frac{6120\sin(\omega\,t)\,\omega+48960\cos(\omega\,t)-48960\,\omega^2\cos(\omega\,t)}{16320-32385\,\omega^2+16320\,\omega^4}$$

```
> steady:=simplify(remove(has,%,exp));
```

$$steady := -\frac{24\left(-\sin(\omega\,t)\,\omega-8\cos(\omega\,t)+8\,\omega^2\cos(\omega\,t)\right)}{64-127\,\omega^2+64\,\omega^4}$$

We now write the steady state in the phase amplitude form.

```
> s0:=simplify(subs(t=0,steady));
```

$$s0 := -\frac{192\left(-1+\omega^2\right)}{64-127\,\omega^2+64\,\omega^4}$$

```
> phase_amp:=A*cos(omega*t-delta);
```

$$phase\_amp := A\cos(\omega\,t-\delta)$$

```
> p0:=simplify(subs(t=0,phase_amp));
```

$$p0 := A\cos(\delta)$$

```
> s1:=simplify(subs(t=0,diff(steady,t)));
```

$$s1 := \frac{24\,\omega^2}{64 - 127\,\omega^2 + 64\,\omega^4}$$

```
> p1:=simplify(subs(t=0,diff(phase_amp,t)));
```

$$p1 := A\sin(\delta)\,\omega$$

```
> solve({p0=s0,p1=s1},{A,delta});
```

$$\left\{\delta = \arctan\left(\omega\,\mathrm{RootOf}\left(-1 + \left(64 - 127\,\omega^2 + 64\,\omega^4\right)\_Z^2, label=\_L12\right), \left(8 - 8\,\omega^2\right)\mathrm{RootOf}\left(-1 + \left(64 - 127\,\omega^2 + 64\,\omega^4\right)\_Z^2, label=\_L12\right)\right), A = 24\,\mathrm{RootOf}\left(-1 + \left(64 - 127\,\omega^2 + 64\,\omega^4\right)\_Z^2, label=\_L12\right)\right\}$$

```
> allvalues(%);
```

$$\left\{A = 24\sqrt{\frac{1}{64 - 127\,\omega^2 + 64\,\omega^4}}, \delta = \arctan\left(\omega\sqrt{\frac{1}{64 - 127\,\omega^2 + 64\,\omega^4}}, \left(8 - 8\,\omega^2\right)\sqrt{\frac{1}{64 - 127\,\omega^2 + 64\,\omega^4}}\right)\right\}, \left\{\delta = \arctan\left(-\omega\sqrt{\frac{1}{64 - 127\,\omega^2 + 64\,\omega^4}}, -\left(8 - 8\,\omega^2\right)\sqrt{\frac{1}{64 - 127\,\omega^2 + 64\,\omega^4}}\right), A = -24\sqrt{\frac{1}{64 - 127\,\omega^2 + 64\,\omega^4}}\right\}$$

```
> assign(%[1]);
```

We take a value of δ corresponding to a positive value for A. We now compute the phase lag for different ω.

```
> evalf(subs(omega=1,delta));
```

$$1.570796327$$

```
> evalf(subs(omega=2,delta));
```

$$3.058451422$$

```
> evalf(subs(omega=3,delta));
```

$$3.094751941$$

So the phase lag is approximately 1.571 when $\omega = 1$, 3.058 when $\omega = 2$, and 3.095 when $\omega = 3$.

```
> plot(delta,omega=0.1..10);
```



We also determine the amplitude of the steady state as a function of the forcing frequency ω.

```
> A;
```

$$24\sqrt{\frac{1}{64-127\,\omega^2+64\,\omega^4}}$$

```
> plot(A,omega=0.1..10);
```



We finally find the value of ω for which the amplitude is maximal.

```
> Aprime:=diff(A,omega);
```

$$Aprime := -\frac{12\left(-254\,\omega+256\,\omega^3\right)}{\sqrt{\frac{1}{64-127\,\omega^2+64\,\omega^4}}\left(64-127\,\omega^2+64\,\omega^4\right)^2}$$

```
> solve(Aprime=0,omega);
```

$$0,\ \frac{1}{16}\sqrt{254},\ -\frac{1}{16}\sqrt{254}$$

```
> evalf(%);
```

$$0., 0.9960860906, -0.9960860906$$

The maximum amplitude occurs when ω is approximately 0.9961. This value is close to 1 at which the corresponding undamped oscillator exhibits resonance. Also not that we exclude $\omega = 0$ since then the forcing is constant.

```
> subs(omega=.9960860906,A);
```

$$24.04701302$$

Also note that unlike resonance the amplitude for this ω is still finite (approximately 24.05).